
В ячейках рабочего листа могут находиться Текст, Константы и Формулы.
Типы данных. Тип определяет объем выделяемой под данные памяти и возможные операции с ними. Опишем основные типы данных Excel.

Целые числа, 25  -4286 Числа, набранные в круглых скобках, трактуются как отрицательные.

Вещественные числа 24,45 (разделитель ‑ запятая) 1,0Е-5 (Е="умножить на 10 в степени")

Внимание! Excel использует стандарты представления даты, времени, валюты, списков, установленные программой Язык и стандарты, входящей в состав Windows. Запустить ее можно, выбрав в меню Пуск пункты Настройка, затем Панель управления.

Дроби: 06/5 (ноль нужен, чтобы отличить дроби от дат)

Для ввода процентов после числа набирается символ %. Если вводимое число является денежным значением, то в конце набирается р. (рубли). Долларовая константа вводится в виде $100.

Если вводимая числовая константа не входит в ячейку по ширине, то она отображается на экране в виде ####. В этом случае ширину столбца необходимо увеличить. 
Дата и Время. Ввод даты, например, 29 октября 1999 года можно провести, набрав на клавиатуре 29/10/99. В памяти компьютера эта дата хранится в виде количества дней между текущей и базовой датами.

Ввод времени выполняется в виде 15:31 или 12:10:00

Ввод текста, состоящего только из цифр: перед таким текстом ставится апостроф. Например, '1234
Создание рядов

Схема ввода одного и того же значения (число, текст, дата, время) во все ячейки диапазона:
1. выделить диапазон;
2. набрать на клавиатуре значение или формулу (без фиксации ввода);
3. при нажатой клавише Ctrl нажать Enter.
Схема ввода одного и того же значения или формулы в часть столбца или строки:

1. ввести в ячейку значение или формулу, зафиксировав ввод;
2. установить указатель мыши на маркере заполнения ячейки и растянуть в нужном направлении при нажатой левой клавише.

Маркер заполнения ячейки - маленький прямоугольник в правом нижнем углу клетки.

Схема ввода числовых значений по типу прогрессии: 

1. ввести в две соседние ячейки два первых элемента прогрессии;
2. выделить эти ячейки;
установить указатель мыши на маркере заполнения выделенных ячеек и растянуть в нужном направлении при нажатой левой клавише.
Данные, сохраненные в ячейках могут отображаться в одном из нескольких форматах. Выбрать формат представления данных и способ оформления ячейки можно в диалоговом окне Формат ячеек. Вызвать его можно нажатием Ctrl+1 или выбором пункта Ячейки… в меню Формат.
Формулы. Все формулы в Excel должны начинаться с символа =.

До фиксации ввода формула отображается в ячейке и в строке формул. После нажатия Enter в ячейке появится вычисленное по формуле значение, а строка ввода очистится.

При вычислении значения по формуле в первую очередь вычисляются выражения внутри круглых скобок. Если скобок нет, то порядок выполнения операций следующий:

1. вычисляются значения функций ;
2. операция возведения в степень (знак операции ^ );
3. операции умножения и деления (знаки операции *, /); 
4. операции сложения и вычитания (знаки операций +,-).
В качестве операндов формула может содержать числа, ссылки (адреса ячеек), функции.

Примеры формул:
= 2*5^ 3+4
=A1+A2
=A1+Cos (5,282)

Значение формулы зависит от содержимого ячеек, на которые указывают ссылки, и оно изменяется при изменении содержимого этих ячеек.

Функции. 
Функция - это заранее определенная формула. Функция имеет имя и аргументы, заключенные в круглые скобки. Аргументы отделяются друг от друга символом ; В качестве аргументов можно использовать другие функции (если они работают с тем же типом данных), константы, адреса ячеек и диапазоны ячеек. Диапазоном называется группа ячеек, образующих прямоугольник. Диапазон обозначается с помощью ячейки в левом верхнем углу прямоугольника и ячейки в правом нижнем углу прямоугольника. Например, обозначение D4:E7 описывает диапазон ячеек, находящихся на пересечении строк с номерами 4,5,6,7 и столбцов D,E.

=СУММ (A1;C2:C5) 

Эта функция имеет два аргумента. Первый A1, второй - C2:C5. Суммируются числа в ячейках A1,C2,C3.C4.C5.

Кнопка Мастер функций на панели инструментов имеет вид fx  .

Адреса ячеек. В Excel различают два типа адресации ячеек: абсолютная и относительная. Оба типа можно применить в одной ссылке и создать смешанную ссылку.

Относительная ссылка воспринимается программой как указание маршрута к адресуемой ячейки от ячейки, содержащей формулу. При копировании формулы относительные ссылки будут изменены таким образом, что маршрут сохранится. Относительные ссылки используются в Excel по умолчанию.

Абсолютная ссылка задает абсолютные координаты ячейки. При копировании формулы абсолютная ссылка на ячейку не будет изменяться. Абсолютная ссылка задается  путем указания символа доллара перед номером строки и столбца, например   $A$2.

Смешанная ссылка представляет собой комбинацию абсолютной и относительной ссылок, когда для строки  и столбца используются разные способы адресации, например, $A1, B$2. При копировании формулы абсолютная часть ссылки не изменяется.

При выделении ссылки на ячейку в режиме редактирования путем нажатия клавиши F4 можно изменять тип ссылки. Пусть введена ссылка А7,тогда

	Нажатие F4
	Адрес

	Один раз

Два раза

Три раза

Четыре раза
	$A$7

A$7

$A7

A7


Задать ссылку, при вводе формулы, можно непосредственно вводом с клавиатуры или методом указания (щелчком мыши на нужной ячейке). 

Часто в формулах  необходимо задавать ссылки на диапазон ячеек. Для указания диапазона Excel используют три адресных оператора:

оператор диапазона (двоеточие): ссылка адресует все ячейки, расположенные между двумя указанными ячейками, например, =СУММ (А1:В2) – возвращает сумму значений ячеек А1, А2, В1 и В2;

оператор объединения диапазонов (точка с запятой): ссылка охватывает ячейки указанных отдельных диапазонов, например, = СУММ (А1; В1: С2) – возвращает сумму ячеек А1, В1, В2, С1, С2;

оператор пересечения диапазонов (пробел): ссылка охватывает ячейки, входящие в каждый из указанных отдельных диапазонов, например, =СУММ (B2:D2 __ C1:D3) – возвращает сумму ячеек C2  и D2.

Полный формат ссылки:

[Книга]Лист!Ячейка

Мастер функций предназначен для ввода готовых формул в вычисления. Кнопка Мастер функций на панели инструментов имеет вид fx 
Выбор функции. Слева в окне Мастера выбрать категорию функции, справа в алфавитном функции найти имя функции, нажать на кнопку ОК.

Ввод аргументов функции. Если у функции есть аргументы, появится окно ввода аргументов, элементы которого показаны на рисунке.

1 - имя функции, для которой вводятся аргументы;

2  - поля ввода аргументов;

3 - кнопка сворачивания окна ввода. Если окно свернуто, развернуть его можно повторным нажатием на эту кнопку;

4 - текущие значения аргументов и функции;

5 - область описания функции;

6 - кнопка вызова помощи.

Ввести аргументы функции можно следующим образом:

а) набрать вручную необходимые адреса или диапазоны ячеек;

б) отметить нужные ячейки или диапазоны ячеек на рабочем листе. Окно ввода аргументов при этом можно свернуть (кнопка 3) или перетащить в сторону.

Вложение функций. Часто необходимо выполнять вычисления "по цепочке", то есть, в качестве аргументов функций указывать значения, вычисленные по другим, вложенным функциям. Чтобы выполнить вложенные вычисления, удобно использовать выпадающий список Имя слева в строке формул. Покажем это на примере: пусть в ячейке A1 набрано число x=1, а в ячейке A2 нужно вычислить величину ln (sin x). Встав в ячейку A2, вызываем Мастер функций, в категории Математические выбираем функцию LN (натуральный логарифм), нажимаем ОК. Для ввода вложенной функции открываем список Имя, если функция SIN есть в списке, выбираем ее, иначе выбираем элемент Другие функции и находим синус во вновь появившемся окне Мастера функций. Для ввода аргумента синуса щелкаем по ячейке A1. Результат: формула вида =LN(SIN(A1))

Описание основных функций.

Категория Дата и время.

Сегодня() - возвращает текущую дату

Год(дата), Месяц(дата), День(дата), ДеньНед (Дата;2) - соответственно, год, месяц, день, день недели. Аргумент 2 у функции ДеньНед нужен для отсчета дней с понедельника. Пример: =ДеньНед(Сегодня();2) - вывести текущий день недели в ячейке

Категория Математические.

ABS (число) – модуль числа.

ACOS (число) – арккосинус числа. угол определяется в радианах в интервале от 0 до (.

ASIN (число) – арксинус числа. Угол определяется в интервале от – (/2 до (/2.

ATAN (число) – арктангенс  числа. Угол определяется в радианах в диапазоне от -(/2 до (/2.

COS(число) – косинус заданного числа.

EXP(число) – возвращает число е, возведенное в указанную степень.

LN(число) – возвращает натуральный логарифм числа.

LOG(число; основание) – возвращает логарифм числа по заданному основанию.

LOG10 (число) – возвращает десятичный логарифм числа

SIN (число) – возвращает синус заданного числа.

TAN (число) – возвращает тангенс заданного числа.

ГРАДУСЫ (угол) – преобразует радианы в градусы.

ЗНАК (число) – определяет знак числа. 1, если число положительное, 0, если число равно 0, и -1, если число отрицательное.

КОРЕНЬ (число) – возвращает положительное значение квадратного корня.

МОБР (массив) – возвращает обратную матрицу для квадратной матрицы, заданной в массиве. Массив может быть задан как интервал ячеек, например А1:С3, или как массив констант {1;2;3: 4;5;6: 7;8;9} (здесь значения в пределах столбца должны быть разделены двоеточием, в пределах строки – точкой с запятой) или как имя массива или интервала. Ввод матричных формул следует завершать нажатием клавиши CTRL + SHIFT +ENTER.

МОПРЕД (массив) – возвращает определитель квадратной матрицы, заданной в массиве.

МУМНОЖ (массив1; массив2) – возвращает произведение матриц, которые задаются массивами. Результатом является массив с таким же числом строк как массив1, и таким же числом столбцов, как массив2.

ОКРУГЛ (число; число_разрядов) – округляет до указанного числа десятичных разрядов.

ОСТАТ (число; делитель) – возвращает остаток от деления аргумента число на делитель.

ПИ () – возвращает число ( с точностью до 15 цифр.

ПРОИЗВЕД(число1; число2; ...) – перемножает числа, заданные в аргументах и возвращает их произведение.

РАДИАНЫ (угол) – преобразует  градусы в радианы.

СТЕПЕНЬ (число; степень) – возвращает результат возведения в степень;

СУММ (число1; число2; ...) – возвращает сумму всех чисел, входящих в список аргументов;

СУММЕСЛИ (диапазон_просмотра; условие_суммирования; диапазон_суммирования) - просматривает диапазон просмотра, выбирает ячейки, отвечающие условию суммирования, и суммирует значения из диапазона суммирования. Размеры диапазона просмотра и диапазона суммирования должны совпадать.
Расширенные возможности частичного суммирования – Сервис.Мастер.Частичная сумма (если не установлен, Сервис.Надстройки)

СУММКВ (число1; число2; ...) – возвращает сумму квадратов аргументов;

СУММКВРАЗН (массв1; массив2) – возвращает сумму квадратов разностей соответствующих значений в двух массивах;

СУММПРОИЗВ (массив1; массив2; ...) – перемножает соответствующие элементы заданных массивов и возвращает сумму произведений;

ФАКТР (число) – возвращает факториал числа;

ЦЕЛОЕ (число) – округляет число до ближайшего целого;

ЧАСТНОЕ (числитель; знаменатель) – Возвращает частное от деления нацело.

Статистические функции.

МАКС(аргумент1; аргумент2;…) - ищет максимальный из аргументов;

МИН(аргумент1; аргумент2;…) - ищет минимальный из аргументов;

СРЗНАЧ(аргумент1; аргумент2;…) - вычисляет среднее своих аргументов;

СЧЕТЕСЛИ(диапазон; условие) - подсчитывает число аргументов в диапазоне, отвечающих условию

Логические функции служат для выполнения вычислений в зависимости от выполнения некоторого условия. В условиях могут использоваться операции сравнения =, >, <, <>(не равно), >= (больше или равно), <= (меньше или равно).

ЕCЛИ (логическое_выражение; значение1; значение2) - результатом является значение1, если логическое_выражение истинно и значение2 в противном случае. Пример: в ячейке A1 набрано число 30000, а в ячейке B1 формула =ЕСЛИ(A1<20000; 12; 15). Результатом будет число 15, т.к. условие на выполняется.

Функции И, ИЛИ служат для создания сложных условий:

И (логическое_выражение1; логическое_выражение2;....) - возвращает значение “ИСТИНА”, если все аргументы имеют значение “ИСТИНА”, а в противном случае -“ЛОЖЬ”. 

ИЛИ (логическое_выражение1; логическое_выражение2;...) - возвращает значение “ИСТИНА”, если хотя бы один из аргументов имеет значение “ИСТИНА”, а противном случае - “ЛОЖЬ”.

Примеры: =ЕСЛИ (И(A1>=20000;A1<40000);15;18) - вычисленное значение равно 15 при величине A1 от 20000 до 40000 и равно 18 в противном случае.

Можно создавать сложные условия и вложением функций ЕСЛИ. Пример: 

=ЕСЛИ (A1<20000; 12; ЕСЛИ (A1<40000; 15; 18)) - если величина A1 меньше 20000, вычисленное значение равно 12, иначе если она меньше 40000, то результат равен 15, а в противном случае (то есть, А1 больше 40000) ,формула вернет значение 18.

Ссылки и массивы. 
Впр (искомое_значение;диапазон поиска;номер_столбца;режим)

Ищет значение в крайнем левом столбце таблицы и возвращает значение в той же строке из указанного столбца таблицы. Режим=ИСТИНА – таблица упорядочена по 1-му столбцу, ЛОЖЬ – нет

Гпр (искомое_значение;диапазон поиска;номер_строки;режим) Ищет значение в верхней строке таблицы или массива значений и возвращает значение в том же столбце из заданной строки таблицы или массива.

Поискпоз (значение; массив;0) - ищет значение в массиве (0 - ищется точное совпадение, 1 <=, -1 >=) и возвращает номер строки, где оно встретилось;

Просмотр(искомая строка; диапазон просмотра; диапазон результатов) – ищет значение в диапазоне просмотра и возвращает соответствующее ему значение из диапазона результатов

Индекс(массив; номер строки) - выбирает значение из массива по индексу.

ВЫБОР(номер_индекса;значение1;значение2;…) Использует номер_индекса, чтобы выбрать и вернуть значение из списка аргументов. Аргументы могут быть ссылками на интервал.

АДРЕС(номер_строки;номер_столбца;тип_ссылки;стиль_ссылки;имя_листа) - Создает адрес ячейки в виде текста, используя номер строки и номер столбца. Тип ссылки 1 – абсолютный, 4 – относительный. Стиль =ИСТИНА – обычная ссылка, ЛОЖЬ – вида C1K1

Пример: БД телефонной сети. Основная таблица содержит поля - фамилия, адрес, номер телефона, дата последней оплаты. Таблица сортируется по фамилиям. Вторая таблица содержит ту же самую информацию, но упорядочена по номерам телефонов. Разработать две квитанции для расчета оплаты по заданному номеру телефона и по фамилии. В квитанции вводится месяц, включая который производится оплата. Учитывается пеня, если оплата производится после 10 числа текущего  месяца.

На листе Таблица1 абоненты упорядочены по номеру телефона, на листе Таблица 2 – по фамилии. Т.к. расчетные формулы однотипны, ниже приводятся комментарии для первого листа. Номера строк и столбцов листа приведены непосредственно в документе Excel (см. на рис. лист Таблица1):

	a
	b
	c
	d
	e
	f

	
	
	
	
	
	

	
	Карточка для расчета оплаты
	
	
	
	

	
	
	
	
	
	

	
	Номер телефона:
	334466
	
	
	5

	
	Месяц:
	6
	
	
	6

	
	
	
	
	
	7

	
	Фамилия абонента:
	Иванов
	
	
	8

	
	Дата последнего взноса:
	08.05.01
	
	
	9

	
	
	
	
	
	10

	
	Сумма оплаты:
	60
	
	
	11

	
	
	
	
	
	

	
	
	
	
	
	

	
	
	
	
	
	

	
	Сегодня:
	28.01.02
	
	
	15

	
	Текущая оплата помесячно:
	60
	
	
	16

	
	Размер пени:
	6
	
	
	17

	
	
	
	
	
	

	(сортировка по номеру телефона)
	
	
	
	

	Фамилия
	Адрес
	Номер телефона
	Дата последней оплаты
	

	Петров
	Советская, 47-04
	237465
	15.05.01
	
	21

	и т.д.
	…
	…
	…
	
	22


Ячейки C5 и C6 предназначены для ввода исходных данных – номера телефона, который нужно найти в базе и номера месяца, до которого производится оплата (включительно).

В ячейке C8 выводится результат поиска абонента по номеру телефона или значение Н/Д, если абонент не найден. Поиск абонента реализуется с помощью формулы: 

=ПРОСМОТР(C5;C21:C1000;A21:A1000)

Здесь C5 – значение искомой строки (номера телефона), C21:C1000 – диапазон просмотра базы данных, A21:A1000 – диапазон, из соответствующей ячейки которого берется результирующее значение.

Для удобства в ячейке С9 выводится дата последнего взноса платы абонентом. Аналогично, поиск реализуется по формуле

=ПРОСМОТР(C5;C21:C1000;D21:D1000)

Сумма оплаты рассчитывается в ячейке С11 по следующей формуле:

=ЕСЛИ(МЕСЯЦ(C9)>=C6;0;(C6-МЕСЯЦ(C9))*C16+ ЕСЛИ(И(ДЕНЬ(C9)>10; МЕСЯЦ(C9)<=C6); C17;0))

Здесь МЕСЯЦ(C9), ДЕНЬ(C9) – месяц и день, когда абонент последний раз вносил плату, С16 – ячейка со значением помесячной платы, С17 – ячейка со значением пени.

Если месяц, когда абонент вносил плату больше либо равен текущего месяца – результат формулы равен 0. Иначе плата вычисляется по формуле

(Месяц, по который оплачиваем - Текущий месяц)*Помесячная плата + Пеня.

Пеня вычисляется по следующему правилу:

ЕСЛИ(И(ДЕНЬ(C9)>10; МЕСЯЦ(C9)<=C6); C17;0))

Если день, когда абонент последний раз вносил плату, больше 10 и месяц, до которого вносилась плата, не превышает текущего, то начисляется пеня, иначе ее значение равно 0.

Значения ошибки при вычислениях по формулам

Если в ячейке содержится формула, результат которой программа не может правильно определить, то в ячейке будет отображено значение ошибки.

	Значение ошибки
	Причина

	# Дел/0!
	Задано деление на 0

	# Знач!
	Указан неправильный аргумент или неправильный оператор

	# Имя?
	Указано недопустимое имя

	# Н/Д 
	Значение не указано

	# Пусто!
	Задана область пересечения двух диапазонов, которые не пересекаются

	#Ссылка!
	Указана некорректная ссылка

	# Число!
	Ошибка при использовании или получении числа


Имена диапазонов

Любым диапазонам ячеек могут быть присвоены имена , которые в дальнейшем можно указывать вместо адресов ячеек в формулах. Имя может содержать не более 255 символов. Первым символом в имени должна быть буква, символ подчеркивания или обратная косая черта. Кроме этого в именах можно использовать точку, вопросительный знак и цифры. При этом имя не должно совпадать с форматом ссылки (B2). 

Имя диапазона может быть задано непосредственно в поле имени строки формул. Для этого следует выделить диапазон ячеек, поместить курсор в поле имени и ввести имя для выделенного диапазона.

Для вставки имени в формулу необходимо открыть список имен, щелкнув на кнопке со стрелкой справа от поля имени, и выбрать нужное имя. Вставить имя можно непосредственно с клавиатуры или, нажав клавишу F3, или, используя меню Вставка, подменю Имя, пункт Вставить. F5 – вызвать окно «Переход»

Матричные и векторные операции.

Признаком матричной операции, то есть, выполнения вычислений сразу в диапазоне ячеек, служит сочетание клавиш Ctrl+Shift+Enter.

Перемножение двух столбцов: нажать кнопку Автосуммирования; выделить 1-й столбец; нажать клавишу умножения (*); выделить 2-й столбец; нажать Ctrl+Shift+Enter;

Проведение вычислений сразу в диапазоне ячеек: выделить диапазон; в первой ячейке ввести число или формулу; нажать Ctrl+Shift+Enter;

Проведение вычислений, результатом которых является диапазон: выделить область результата; с помощью Мастера функций или вручную ввести формулу; после ввода последнего аргумента нажать Ctrl+Shift+Enter;

Подбор параметра. Значение определенной (целевой) ячейки является результатом вычисления формулы. Эта формула прямо или косвенно ссылается на одну или несколько влияющих ячеек. Функция подбора меняет значение влияющей ячейки так, чтобы получить в целевой ячейке заданную величину.

Пример. С помощью подбора параметра найти корень нелинейного уравнения

x5 - 4x4 + 3x3 - 2x2 + x - 1 = 0, x принадлежит отрезку [-5,5]

Для этого в ячейку B1 введем произвольное значение X (при некоторых начальных значениях процесс решения может и не сойтись), в ячейку B2 введем формулу =B1^5-4*B1^4+3*B1^3-2*B1^2+B1-1 Установив курсор в ячейку B2, обратимся к команде Подбор параметра меню Сервис. В поле “значение” окна диалога введем число 0, в поле “изменяя значение ячейки” – значение B1 и нажмем кнопку ОК. Excel проиллюстрирует результаты подбора параметра в новом окне диалога.

Поиск решения – это нахождение оптимального значения исследуемой функции. Ячейка, в которой находится функция, может ссылаться на изменяемые ячейки, в которых содержатся ее аргументы. При этом для каждого аргумента можно задать ограничения. Для запуска поиска решения нужно встать в целевую ячейку и выбрать команду Поиск решения меню Сервис.

После этого, выбрав в окне диалога критерий оптимизации (минимальное, максимальное или фиксированное значение целевой функции), нужно сослаться на зависимые ячейки и ввести ограничения в соответствующих полях окна диалога. Ограничения указываются в виде

Зависимая_ячейка Знак Выражение,

где Знак может быть <=, =, >= или ограничение до целого числа, если задача целочисленна.

При этом начальные значения зависимых ячеек должны быть таковы, чтобы численный метод оптимизации сходился.

Пример: определить длины сторон a,b,h прямоугольного бака заданного объема V, минимизируя длину сварного шва, которая вычисляется по формуле: L=2(a+2b)+h. Математическая модель этой задачи вместе с ограничениями такова: L→min – оптимизируемая функция; V=const; a,b,h>0 – ограничения.

Решение: Введем начальные значения зависимых переменных а=1, b=1, c=1 в ячейки B3, C3, D3, а ограничение для них – число 0 – в ячейку B4. Зависимую переменную  V=a*b*h введем в ячейку C7 в виде формулы =B3*C3*D3, а ограничение на нее в виде значения 2 – в ячейку E7. Целевую формулу для вычисления L в виде =2*(B3+2*C3)+D3 введем в ячейку C8. После этого в окне Поиск решения укажем $C$8 в качестве целевой ячейки, выберем поиск минимального значения, в поле «Изменяя ячейки» укажем ссылку $B$3:$D$3, в поле ограничения введем ограничения вида $B$3>=$B$4, $C$3>=$B$4,$D$3>=$B$4, $C$7=$E$7. Найденное решение должно быть таким: A=1,26, B=0,63, C=2,52.

